# Maintainer: Antonio Rojas <arojas@archlinux.org>
# Contributor: Rémy Oudompheng <oudomphe@clipper.ens.fr>

pkgname=symmetrica
pkgver=3.0.1
pkgrel=1
pkgdesc="Algorithms for symmetric groups and classical groups"
arch=(loong64 x86_64)
url="http://www.algorithm.uni-bayreuth.de/en/research/SYMMETRICA/"
depends=(glibc)
license=(custom)
source=("https://gitlab.com/sagemath/symmetrica/-/archive/$pkgver/$pkgname-$pkgver.tar.gz")
sha256sums=('6e0fea5708d2deed8369ce4c7c2123a23a1c942d8cdf39c1f81258aec3ac04ec')

prepare() {
  cd $pkgname-$pkgver
  autoreconf -fi
}

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
  install -Dm644 LICENSE -t "$pkgdir"/usr/share/licenses/symmetrica
}
